

## ESTIMATES OF SOME FUNCTIONS OVER PRIMES WITHOUT THE RIEMANN HYPOTHESIS

PIERRE DUSART

e-mail: pierre.dusart@unilim.fr

Revision : 26/02/2018

Thomas Morrill tried to reproduce Table 1 and asked me how to calculate this table. I found that there are two typing errors in the published theorem. The correct version is the following:

**Theorem 3.5.** *Let  $m \in \mathbb{N}$ ,  $m \geq 2$ ,  $\delta > 0$ , and the pair  $(a, b)$  takes values  $(1, 1 + \delta)$  or  $(1 - \delta, 1)$ . Let  $H, T_0, T_1, R, \sigma_0, c_1$  satisfy (3.1). Let  $b_0$  be a positive constant. Then for all  $x \geq e^{b_0}$ ,*

$$\begin{aligned} \frac{|\psi(x) - x|}{x} \leq & \max_{(a,b)} \left\{ \frac{2M(a, b, m)}{\delta^m} \left( B_5 + B_3(e^{-(1-\sigma_0)b_0} + e^{-\sigma_0 b_0}) + B_4 e^{-(1-\frac{1}{R \log H})b_0} \right) \right. \\ & + 2 \left( M(a, b, 0)B_1 + \frac{M(a, b, m)}{\delta^m} B_2 \right) e^{-b_0/2} + \frac{\delta}{2} \\ & \left. + \ln(2\pi)e^{-b_0} + \frac{M(a, b, 0)}{2} e^{-3b_0} \right\} \end{aligned} \quad (3.3)$$

```

/***** Code gp/pari *****/
T_0 = 1132490.982
sum_bound = 11.6377324
R = 5.573412
H = 2445999556030
a_1 = 0.137
a_2 = 0.443
a_3 = 1.588
r(T) = a_1*log(T) + a_2*log(log(T)) + a_3
q(y) = (a_1*log(y) + a_2)/(y*log(y)*log(y/2/Pi))
c(sigma_) = log(1 + 9.8/2/H*(3*H)^(8*(1-sigma_)/3)*log(H)^(5-2*sigma_)) + 103*log(H)^2/2/H

B_1(T) = sum_bound + (1/2/Pi + q(T_0))*(log(T/T_0)*log(sqrt(T*T_0)/2/Pi)) + 2*r(T_0)/T_0
B_2(m,T) = (1/2/Pi + q(T))*((1+m*log(T/2/Pi))/(m^2*T^m) - (1+m*log(H/2/Pi))/(m^2*H^m))
           + 2*r(T)/T^(m+1)
B_3(m) = (1/2/Pi + q(H))*(1+m*log(H/2/Pi))/(m^2*H^m) + 2*r(H)/H^(m+1)
B_4(m,sigma_) = c(sigma_)*(1+1/m)/H^m

```

2

```
M(a,b,m) = factorial(2*m+1)/factorial(m)
           *intnum(u= 0, 1,abs(pollegendre(m,1 - 2*u))*((b-a)*u+a)^(m+1) )

B_5(b_0, m,sigma_) = c(sigma_)*(1 + R/2/b_0*log(H)^2/(m*R/b_0*log(H)^2-1))
                    *exp(-b_0/R/log(H))/H^m

eps(b_0, a,b,m,sigma_,delta,T) =
    2*M(a,b,m)/delta^m*(B_5(b_0, m,sigma_) +B_3(m)*(exp((sigma_-1)*b_0)
    + exp(-b_0*sigma_)) + B_4(m,sigma_)*exp(-(1-1/R/log(H))*b_0))
    + 2*(M(a,b,0)*B_1(T) +M(a,b,m)/delta^m*B_2(m,T))*exp(-b_0/2)
    + delta/2 +log(2*Pi)*exp(-b_0) + M(a,b,0)/2*exp(-3*b_0)

max_eps(b_0,sigma_,m,delta,T) =max(eps(b_0,1,1+delta,m,sigma_,delta,T),
    eps(b_0,1-delta,1,m,sigma_,delta,T))

max_eps(45, 0.87, 3, 2.721E-9, 2228096512)
/* = 1.2243692462327181678372667055171613298 E-8 */

max_eps(1000, 0.97, 9, 1.574E-11, 1012519261279)
/* = 8.7424255202857532866461985874516705831 E-12 */
```